Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematical Sciences
goutzicj@oneonta.edu

# Lesson 4

## Separation of Variables

# Initializations

```
> restart;
  with(plots):
  with(oneonta):
```

# 4.1 Separation of Variables

## Examples

### *Example 4.1.1*

**Find the general solution of the differential equation**

$$\frac{dy}{dt} = \frac{t}{y^2}$$

**and plot some of the solution curves.**

**Solution**
Code the differential equation and separate the variables.

```
> deq:=diff(y(t), t)=t/y(t)^2;
```

$$deq := \frac{d}{dt}\, y(t) = \frac{t}{y(t)^2} \qquad \textbf{(2.1.1.1)}$$

```
> deq1:=deq*y(t)^2;
```

$$deq1 := y(t)^2 \left(\frac{d}{dt}\, y(t)\right) = t \qquad \textbf{(2.1.1.2)}$$

Integrate left and right hand side by mapping the integration routine **int** to the equation. Do not forget to include an integration constant.

```
> sol1:=map(int, deq1, t)+(0=c);
```

$$sol1 := \frac{1}{3}\, y(t)^3 = \frac{1}{2}\, t^2 + c \qquad \textbf{(2.1.1.3)}$$

Solve for $y(t)$.

```
> sol2:=solve(sol1, y(t));
```

$$sol2 := \frac{1}{2}\left(12\, t^2 + 24\, c\right)^{1/3}, -\frac{1}{4}\left(12\, t^2 + 24\, c\right)^{1/3} + \frac{1}{4}\, \mathrm{I}\sqrt{3}\left(12\, t^2 + 24\, c\right)^{1/3}, -\frac{1}{4}\left(12\, t^2 + 24\, c\right)^{1/3} - \frac{1}{4}\, \mathrm{I}\sqrt{3}\left(12\, t^2 + 24\, c\right)^{1/3} \qquad \textbf{(2.1.1.4)}$$

Extract the real solution.

```
> sol3:=sol2[1];
```

$$sol3 := \frac{1}{2}\left(12\,t^2 + 24\,c\right)^{1/3} \qquad \textbf{(2.1.1.5)}$$

Different constants $c$ correspond to different solutions of the differential equation. For example we will take

$$c = -5, -4, ..., 4, 5$$

```
> curves:=[seq(sol3, c=-5..5)];
```

$$curves := \left[\frac{1}{2}\left(12\,t^2 - 120\right)^{1/3}, \frac{1}{2}\left(12\,t^2 - 96\right)^{1/3}, \frac{1}{2}\left(12\,t^2 - 72\right)^{1/3}, \frac{1}{2}\left(12\,t^2 - 48\right)^{1/3}, \frac{1}{2}\left(12\,t^2 - 24\right)^{1/3}, \frac{1}{2}\,12^{1/3}\left(t^2\right)^{1/3}, \frac{1}{2}\left(12\,t^2 + 24\right)^{1/3}, \frac{1}{2}\left(12\,t^2 + 48\right)^{1/3}, \frac{1}{2}\left(12\,t^2 + 72\right)^{1/3}, \frac{1}{2}\left(12\,t^2 + 96\right)^{1/3}, \frac{1}{2}\left(12\,t^2 + 120\right)^{1/3}\right] \qquad \textbf{(2.1.1.6)}$$

In Maple, odd roots of negative real numbers will by default evaluate to a complex number. In order to force the software to generate the real function values, we use the **realroots** command in the **oneonta** package.

```
> plot(realroots(curves), t=-5..5, color=red);
```




Of course we have to be careful with values of $t$ for which $y(t) = 0$, because if $y(t) = 0$, the right hand side of the differential equation is undefined. These details will be addressed later in the course.

*Example 4.1.2*

**Compute the solution of the initial value problem**

$$\frac{dy}{dt} = \frac{y}{1 + y^2}, \quad y(0) = 2$$

**Plot the result.**

**Solution**

Code the equation, separate the variables and integrate.

```
> deq:=diff(y(t), t)=y(t)/(1+y(t)^2);
```

$$deq := \frac{d}{dt}\, y(t) = \frac{y(t)}{1 + y(t)^2} \qquad \textbf{(2.1.2.1)}$$

```
> deq1:=deq/rhs(deq);
```

$$deq1 := \frac{\left(1 + y(t)^2\right)\left(\frac{d}{dt}\, y(t)\right)}{y(t)} = 1 \qquad \textbf{(2.1.2.2)}$$

```
> sol1:=map(int, deq1, t)+(0=c);
```

$$sol1 := \frac{1}{2}\, y(t)^2 + \ln(y(t)) = t + c \qquad \textbf{(2.1.2.3)}$$

Observe that it is not possible to express $y$ in terms of elementary functions of $t$. We therefore apply the initial condition to this implicit representation.

```
> val_c:=isolate(subs({t=0, y(t)=2}, sol1), c);
```

$$val\_c := c = 2 + \ln(2) \qquad \textbf{(2.1.2.4)}$$

```
> sol2:=subs(val_c, sol1);
```

$$sol2 := \frac{1}{2}\, y(t)^2 + \ln(y(t)) = t + 2 + \ln(2) \qquad \textbf{(2.1.2.5)}$$

In order to be able to use the **implicitplot** routine to plot this implicit solution function, we replace $y(t)$ by $y$.

```
> sol3:=subs(y(t)=y, sol2);
```

$$sol3 := \frac{1}{2}\, y^2 + \ln(y) = t + 2 + \ln(2) \qquad \textbf{(2.1.2.6)}$$

```
> implicitplot(sol3, t=-5..5, y=0..4);
```